Problem:
Score : 400 points

Problem Statement

There are N cities. There are also K roads and L railways, extending between the cities.
The i-th road bidirectionally connects the p_i-th and q_i-th cities, and the i-th railway bidirectionally connects the r_i-th and s_i-th cities.
No two roads connect the same pair of cities. Similarly, no two railways connect the same pair of cities.

We will say city A and B are connected by roads if city B is reachable from city A by traversing some number of roads. Here, any city is considered to be connected to itself by roads.
We will also define connectivity by railways similarly.

For each city, find the number of the cities connected to that city by both roads and railways.

Constraints

2 ≦ N ≦ 2*10^5

1 ≦ K, L≦ 10^5

1 ≦ p_i, q_i, r_i, s_i ≦ N

p_i < q_i

r_i < s_i

When i ≠ j, (p_i, q_i) ≠ (p_j, q_j)

When i ≠ j, (r_i, s_i) ≠ (r_j, s_j)

Input

The input is given from Standard Input in the following format:

N K L
p_1 q_1
:
p_K q_K
r_1 s_1
:
r_L s_L

Output

Print N integers. The i-th of them should represent the number of the cities connected to the i-th city by both roads and railways.

Sample Input 1

4 3 1
1 2
2 3
3 4
2 3

Sample Output 1

1 2 2 1

All the four cities are connected to each other by roads.

By railways, only the second and third cities are connected. Thus, the answers for the cities are 1, 2, 2 and 1, respectively.

Sample Input 2

4 2 2
1 2
2 3
1 4
2 3

Sample Output 2

1 2 2 1

Sample Input 3

7 4 4
1 2
2 3
2 5
6 7
3 5
4 5
3 4
6 7

Sample Output 3

1 1 2 1 2 2 2

Solution in OCaml (4.02.3) (judge verdict: Accepted)

module M = Map.Make(struct type t = int * int let compare = compare end)
let inc k m = try M.add k (1 + M.find k m) m with Not_found -> M.add k 1 m

let init n = Array.make n (-1)
let rec find x p =
  if p.(x) < 0 then x
  else let x' = find p.(x) p in p.(x) <- x'; x'
let unite x y p =
  let x', y' = find x p, find y p in
  if x' = y' then () else begin
    if p.(x') < p.(y') then begin
      p.(x') <- p.(x') + p.(y'); p.(y') <- x';
    end else begin
      p.(y') <- p.(x') + p.(y'); p.(x') <- y';
    end
  end
let size x p = - p.(find x p)

let () =
  Scanf.scanf "%d %d %d" @@ fun n k l ->
  let e0 = Array.init k (fun _ -> Scanf.scanf " %d %d" (fun x y -> x-1, y-1)) in
  let e1 = Array.init l (fun _ -> Scanf.scanf " %d %d" (fun x y -> x-1, y-1)) in
  let uf0, uf1, uf2  = init n, init n, init n in
  e0 |> Array.iter (fun (u, v) -> unite u v uf0);
  e1 |> Array.iter (fun (u, v) -> unite u v uf1);
  let g = Array.init n (fun i -> (find i uf0, find i uf1)) in
  let m = Array.fold_left (fun m (i, j) -> inc (i, j) m) M.empty g in
  g |> Array.iter (fun p -> M.find p m |> Printf.printf "%d ");
  print_newline ()